
ATSC Standard: 
A/331:2025-06 Amendment No. 2, “User Agent String” 

1. OVERVIEW 

1.1 Definition 
An Amendment is generated to document an enhancement, an addition or a deletion of 
functionality to previously agreed technical provisions in an existing ATSC document. 
Amendments shall be published as attachments to the original ATSC document. Distribution by 
ATSC of existing documents shall include any approved Amendments. 

1.2 Scope 
This document describes corrections to the User Agent string specification in Sec. 7.1.2.4. 

1.3 Rationale for Changes 
The changes described in this document are being proposed because the normative grammar 
(which has precedence over the rest of the text) has an error: the regular user-agent string described 
as “HTML” in Sec. 7.1.2.4 is described incompletely in the grammar. 

Note in particular that the syntax is described as: 

“ATSC3/”<yyyy>[-mm] “(“<capabilities>; <vendorName>; <modelName>; 
<softwareVersion>; <familyName>;[ hardwareVersion]”)” <HTML> 

But the start rule of the grammar is 

ATSCUser-Agent = ATSCproduct RWS ATSCcomment 

The “<HTML>” aspect of the syntax is not present in the start rule. 

1.4 Compatibility Considerations 
The changes described in this document are backward-compatible relative to the currently 
published version of the standard to which this Amendment pertains and any previously approved 
Amendments for that standard. In particular, verification of an ATSC 3 user-agent field should 
terminate after the ATSC3 product. 

Note that the ATSC3 product field in the example in A/331 does successfully validate against 
the grammar; validation fails immediately following “5.234)”. 

ATSC3/2022-06 (50B 512 &; TVco; Model55; R4.2; com.tvco.55; 5.234) 
Mozilla/5.0 (Windows NT 10.0; Win64; x64) AppleWebKit/537.36 (KHTML, 

like Gecko) Chrome/75.0.3770.142 Safari/537.36 

2. LIST OF CHANGES 
Change instructions are given below in italics. Unless otherwise noted, inserted text, tables, and 
drawings are shown in blue; deletions of existing text are shown in red strikeout. The text “[ref]” 
indicates that a cross reference to a cited referenced document should be inserted. 

2.1 Change Instructions 
In Section 7.1.2.4, revise as shown: 

7.1.2.4 HTTP User-Agent Header Field 
In order for HTTP servers to customize the delivery of apps, HTTP requests from the receiver on 
behalf of a Broadcast Application, including the launch URL request, shall provide a User-Agent 
header field as defined in HTTP [32] and this section. 

The syntax shall conform to HTTP [32] Section 5.5.3 and the syntax defined below: 

“ATSC3/”<yyyy>[-mm] “(“<capabilities>; <vendorName>; <modelName>; 
<softwareVersion>; <familyName>;[ hardwareVersion]”)” <HTML> 

where: 
yyyy shall be the required year of publication of A/300 
mm shall be the optional month of publication of A/300, which is required if it is the second or more 

publication in that year 
capabilities shall be a required string conforming to the syntax of sa:Capabilities as defined in 

A/332 [5] Section 5.2.2.3.3 
vendorName shall be a required string of the Receiver vendor’s name 
modelName shall be a required string of the Receiver vendor’s model name (vendor-dependent) 
softwareVersion shall be a required string of the Receiver vendor’s software version (vendor-

dependent) 
familyName shall be a required string of the Receiver vendor’s family name, e.g. org,vendor.x 

(vendor-dependent) 
hardwareVersion shall be an optional string of the Receiver vendor’s model name (vendor-

dependent) 
HTML shall be the receiver’s HTML User-Agent information 

Although the full syntax of “capabilities” is permitted, it is expected that supported capabilities 
are ANDed together (e.g., “509 50a &”), and it is expected that the OR (“|”) is not used. 

The ABNF [23] below is authoritative over the pseudocode and description above. 

; !syntax("abnf") 

ATSCUser-Agent = ATSCproduct RWS ATSCcomment RWS *User-Agent 

ATSCproduct = "ATSC3/" date 

ATSCcomment = "(" SAcapabilities vendorName modelName softwareVersion 
familyName [hardwareVersion] ")" 

date  = year ["-" month] 

year  = 4DIGIT 

month  = 2DIGIT 

SAcapabilities = capabilities 

vendorName = commentBody 

modelName = commentBody 

softwareVersion = commentBody 

familyName = commentBody 

hardwareVersion = commentBody 

commentBody = ";" RWS 1*(ATSCctextNSP / quoted-pair) 

ATSCctext = HTAB / SP / ATSCctextNSP 

ATSCctextNSP = %x21-27 / %x2A-3A / %x3C-5B / %x5D-7E / obs-text ;except 
semicolon 

; A/332 

ccode  = 1*4HEXDIG 

string-code = ubyte "=" 1*UTF8-char 

ubyte  = 1*3DIGIT 

boperator = "&" / "|" 

expr  = ccode / string-code 

    / expr WSP expr WSP boperator 

capabilities = expr 

 

; RFC 7231 

User-Agent = product *( RWS ( product / comment ) ) 

product         = token ["/" product-version] 

product-version = token 

 

; RFC 7230 

token  = 1*tchar 

tchar  = "!" / "#" / "$" / "%" / "&" / "'" / "*" 

                    / "+" / "-" / "." / "^" / "_" / "`" / "|" / "~" 

                    / DIGIT / ALPHA 

comment  = "(" *( ctext / quoted-pair / comment ) ")"  

ctext  = HTAB / SP / %x21-27 / %x2A-5B / %x5D-7E / obs-text 

quoted-pair = "\" ( HTAB / SP / VCHAR / obs-text ) 

obs-text  = %x80-FF 

RWS   = 1*( SP / HTAB ) 

 

; RFC 5234 

ALPHA  = %x41-5A / %x61-7A   ; A-Z / a-z 

DIGIT  = %x30-39 

HEXDIG  =  DIGIT / "A" / "B" / "C" / "D" / "E" / "F" 

HTAB  = %x09 

SP   = " " 

VCHAR  = %x21-7E 

WSP   = " " / %x09 

 

;RFC 3629 

UTF8-char   = UTF8-1 / UTF8-2 / UTF8-3 / UTF8-4 

UTF8-1      = %x00-7F 

UTF8-2      = %xC2-DF UTF8-tail 

UTF8-3      = %xE0 %xA0-BF UTF8-tail / %xE1-EC 2( UTF8-tail ) / 

                 %xED %x80-9F UTF8-tail / %xEE-EF 2( UTF8-tail ) 

UTF8-4      = %xF0 %x90-BF 2( UTF8-tail ) / %xF1-F3 3( UTF8-tail ) / 

                 %xF4 %x80-8F 2( UTF8-tail ) 

UTF8-tail   = %x80-BF 

An example for a receiver conforming to A/300:2022 that signals SHEVC and AC-4 with 
“HTML” strings from a popular web browser is as follows: 

ATSC3/2022-06 (50B 512 &; TVco; Model55; R4.2; com.tvco.55; 5.234) 
Mozilla/5.0 (Windows NT 10.0; Win64; x64) AppleWebKit/537.36 

(KHTML, like Gecko) Chrome/75.0.3770.142 Safari/537.36 

– End of Document – 

